

Initiation à l'économétrie des données de panel

François LEGENDRE

Centre d'Études de l'Emploi

Mise en œuvre avec le système Sas

FIG. 1 – La feuille Excel pour la valeur ajoutée brute

2.202 Valeur ajoutée par branche en volume (prix chaînés, base 2000)		1978	1979	1980	1981	1982	1983	1984
DA	Agriculture, sylviculture, pêche	22,3	24,7	24,7	24,6	28,5	25,3	2
DB	Industrie (= EB à EG)	166,7	168,2	169,2	167,1	165,2	166,0	16
EB	Industries agricoles et alimentaires	21,1	21,9	22,9	23,7	23,3	23,1	2
FB1	Industries de la viande et du lait	6,0	6,5	7,1	7,5	7,1	7,5	
FB2	Autres industries agricoles et alimentaires	15,2	15,4	15,9	16,2	16,2	15,6	1
EC	Industries des biens de consommation	28,0	28,9	29,4	29,4	30,8	31,2	3
FC1	Habillement, cuir	6,9	7,1	7,1	7,1	7,4	7,5	
FC2	Edition, imprimerie, reproduction	9,9	9,9	10,0	9,9	10,3	10,4	1
FC3	Pharmacie, parfumerie et entretien	4,7	4,7	4,8	4,9	5,0	5,3	
FC4	Industries des équipements du foyer	6,6	7,1	7,4	7,4	8,0	7,9	
ED	Industrie automobile	10,3	9,8	9,7	8,9	9,4	9,8	
EE	Industries des biens d'équipement	35,4	35,1	34,9	34,6	32,9	32,1	2
FE1	Construction navale, aéronautique et ferroviaire	5,9	5,8	6,1	7,4	7,9	8,0	
FE2	Industries des biens d'équipement mécaniques	24,6	24,1	23,3	21,7	20,2	18,7	1
FE3	Industries des équipements électriques et électroniques	5,5	5,8	6,1	6,4	6,0	6,6	
EF	Industries des biens intermédiaires	46,9	46,9	47,0	46,8	47,7	49,7	5

Listing 1 – Lecture des feuilles Excel

```
1 %let secteurs = 'ED', 'FB1', 'FB2', 'FC1', 'FC2', 'FC3', 'FC4',  
  'FE1', 'FE2', 'FE3', 'FF1', 'FF2', 'FF3', 'FF4', 'FF5', 'FF6',  
  'FG1', 'FG2' ;  
2 %macro lecture(classeur, secteur, ident) ;  
3   proc import file = &classeur dbms = excel5 out = &ident  
4     replace ;  
5   run ;  
6   data &ident ;  
7     set &ident ;  
8     secteur = &secteur ;  
9     if secteur in (&secteurs) ;  
10    ident = "&ident" ;  
11  run ;  
12 %mend ;  
13 %lecture('c:\xyz\t_2202.xls', z2_202va, q) ;  
14 %lecture('c:\xyz\t_2604.xls', z2_604ca, k) ;  
15 %lecture('c:\xyz\t_2205.xls', z2_205em, l) ;
```

Listing 2 – Construction des variables

```

15 data table ;
16     set q k l ;
17 run ;
18 proc sort data = table out = table ;
19     by secteur ;
20 run ;
21 proc transpose data = table out = table ;
22     by secteur ;
23     id ident ;
24     var var2 - var29 ;
25 run ;
26 data table ;
27     set table ;
28     by secteur ;
29     if first.secteur then num_secteur + 1 ;
30     i = secteur ;
31     t = substr(_label_, 4, 2) - 2 + 1978 ;
32     lq = log(q) ;
33     lk = log(k) ;
34     ll = log(l) ;
35     array dt dt1978 - dt2005 ;
36     do indice = 1 to dim(dt) ;
37         dt{indice} = 0 ;
38     end ;
39     dt{t-1978+1} = 1 ;
40     array di di1 - di18 ;
41     do indice = 1 to dim(di) ;
42         di{indice} = 0 ;
43     end ;
44     di{num_secteur} = 1 ;
45     keep q k l i t lq lk ll dt1978 - dt2005 di1 - di18 ;
46 run ;

```

Listing 3 – Estimation dans la dimension totale

```

47 proc reg data = table ;
48     model lq = lk ll dt1978 - dt2005 / noint ;
49 run ;

```

Listing 4 – Estimation dans la dimension intra-individuelle - I

```

50 proc reg data = table ;
51     model lq = lk ll di1 - di18 dt1978 - dt2004 / noint ;
52 run ;

```

Listing 5 – Construction des variables « moyennées »

```

53 proc means data = table noprint ;
54     var lq lk ll dt1978 – dt2005 ;
55     by i ;
56     output out = between mean = lqb lkb llb dtb1978 – dtb2005 ;
57 run ;
58 data within ;
59     merge table between ;
60     by i ;
61     array w lqw lkw llw dtw1978 – dtw2005 ;
62     array b lqb lkb llb dtb1978 – dtb2005 ;
63     array o lq lk ll dt1978 – dt2005 ;
64     do indice = 1 to dim(w) ;
65         w{indice} = o{indice} – b{indice} ;
66     end ;
67     keep i lqw lkw llw dtw1978 – dtw2005 ;
68 run ;

```

Listing 6 – Décomposition de la variance

```

69 proc univariate data = table ;
70     var lq lk ll ;
71 run ;
72 proc univariate data = within ;
73     var lqw lkw llw ;
74 run ;
75 proc univariate data = between ;
76     var lqb lkb llb ;
77 run ;

```

Listing 7 – Estimation dans la dimension intra-individuelle - II

```

78 proc reg data = within outest = outestWithin ;
79     model lqw = lkw llw dtw1978 – dtw2004 / noint ;
80 run ;

```

Listing 8 – Estimation dans la dimension inter-individuelle

```

81 proc reg data = between outest = outestBetween ;
82     model lqb = lkb llb ;
83 run ;

```

Listing 9 – Construction des variables quasi « moyennées »

```

84 proc contents data = between out = contentsBetween noprint ;
85 run ;
86 proc contents data = table out = contentsTable noprint ;
87 run ;
88 data mcqg ;
89     if _n_ eq 1 then do ;
90         set contentsBetween ; n = nobs ;
91         set contentsTable ; t = nobs / n ;
92         set outestWithin ; scr_w = _rmse_*_rmse_ * n * t ;
93         set outestBetween ; scr_b = _rmse_*_rmse_ * n * t ;
94         stheta = sqrt( (scr_w/(n*t-n)) / (scr_b/n)) ;
95         retain stheta ;
96     end ;
97     merge table between (drop = _type_) ;
98         by i ;
99     array g lqg lkg llg dtg1978 – dtg2005 ;
100    array b lqb lkb llb dtb1978 – dtb2005 ;
101    array o lq lk ll dt1978 – dt2005 ;
102    do indice = 1 to dim(g) ;
103        g{indice} = o{indice} – (1–stheta)*b{indice} ;
104    end ;
105    keep i lqg lkg llg dtg1978 – dtg2005 ;
106 run ;

```

Listing 10 – Estimation des moindres carrés quasi généralisés

```

107 proc reg data = mcqg ;
108     model lqg = lkg llg dtg1978 – dtg2005 / noint ;
109 run ;

```